
Blatt 9: Musterlösung

ACHTUNG: Die Musterlösung ist ein zusätzliches Serviceangebot. Sie erhebt weder Anspruch
auf Vollständigkeit noch auf Korrektheit.

Aufgabe 1: (?) Arithmetik

Aufgabe 2: (?) Bitweise Operationen

Aufgabe 3: (?) Speicherzugriffe

Aufgabe 4: (?) Absolutbetrag

Aufgabe 5: (?) Hammingabstand

Aufgabe 6: Programmieren
Schreiben Sie ein MMIX-Programm, das einen Rechner nach umgekehrter polnischer Notation
implementiert.

Lösungsvorschlag

LOC Data_Segment

stackPtr GREG 0

x0 GREG @

X0 IS @

N IS 1024

LOC #100

charEnd IS 0

charNL IS 10

charPlus IS 43

charMinus IS 45

charStar IS 42

charSlash IS 47

charZero IS 48

charNine IS 57

charSpace IS 32

ptr IS $0

chr IS $1

size IS $2

temp IS $3

haveValue IS $4

value IS $5

Main SET value,0

SET stackPtr,1024

GETA $255,9F

TRAP 0,Fgets,StdIn

BN $255,Error

BZ $255,Exit

SET size,$255

SET ptr,0

SET haveValue,0

1H LDBU chr,x0,ptr

ADD ptr,ptr,1

CMPU temp,chr,charEnd

BZ temp,Main

CMPU temp,chr,charNL

BZ temp,Done

CMPU temp,chr,charZero

BN temp,2F

CMPU temp,chr,charNine

BP temp,2F

SUB temp,chr,48

SET haveValue,1

MUL value,value,10

ADD value,value,temp

JMP 1B

2H BZ haveValue,3F

* SET $6,value

PUSHJ 4,pushValue

SET haveValue,0

SET value,0

3H CMP temp,chr,charSpace

BZ temp,1B

CMP temp,chr,charPlus

BNZ temp,3F

PUSHJ 5,add

JMP 4F

3H CMP temp,chr,charMinus

BNZ temp,3F

PUSHJ 5,subtract

JMP 4F

3H CMP temp,chr,charStar

BNZ temp,3F

PUSHJ 5,multiply

JMP 4F

3H CMP temp,chr,charSlash

BNZ temp,3F

PUSHJ 5,divide

JMP 4F

3H JMP Error

4H CMPU temp,ptr,size

BN temp,1B

JMP Main

Done BZ haveValue,3F

PUSHJ 4,pushValue

3H PUSHJ 6,popValue

SET value,$6

BNN value,1F

GETA $255,Minus

TRAP 0,Fputs,StdOut

NEG value,value

1H SWYM

SET ptr,10

SET temp,0

STBU temp,x0,ptr

2H SUB ptr,ptr,1

DIV value,value,10

GET temp,rR

ADD temp,temp,charZero

STBU temp,x0,ptr

BNZ value,2B

LDA $255,x0,ptr

TRAP 0,Fputs,StdOut

GETA $255,NewLine

TRAP 0,Fputs,StdOut

Exit SET $255,0

TRAP 0,Halt,0

Error SET $255,0

SUB $255,$255,1

TRAP 0,Halt,0

9H OCTA X0,N

Minus BYTE "-",0,0,0

NewLine BYTE 10,0,0,0

pushValue STOU $0,x0,stackPtr

ADD stackPtr,stackPtr,8

POP 0,0

popValue SUB stackPtr,stackPtr,8

LDOU $0,x0,stackPtr

POP 1,0

add GET $0,rJ

PUSHJ 1,popValue

PUSHJ 2,popValue

ADD $2,$2,$1

PUSHJ 1,pushValue

PUT rJ,$0

POP 0,0

subtract GET $0,rJ

PUSHJ 1,popValue

PUSHJ 2,popValue

SUB $2,$2,$1

PUSHJ 1,pushValue

PUT rJ,$0

POP 0,0

multiply GET $0,rJ

PUSHJ 1,popValue

PUSHJ 2,popValue

MUL $2,$2,$1

PUSHJ 1,pushValue

PUT rJ,$0

POP 0,0

divide GET $0,rJ

PUSHJ 1,popValue

PUSHJ 2,popValue

DIV $2,$2,$1

PUSHJ 1,pushValue

PUT rJ,$0

POP 0,0

